# $Id: PKGBUILD 157779 2012-04-29 10:55:02Z allan $
# Maintainer: Роман Кирилич (Roman Kyrylych) <roman@archlinux.org>
# Contributor: Aaron Griffin <aaron@archlinux.org>
# Contributor: Vinay S Shastry <vinayshastry@gmail.com>

pkgname=yasm
pkgver=1.2.0
pkgrel=2
pkgdesc="A rewrite of NASM to allow for multiple syntax supported (NASM, TASM, GAS, etc.)"
arch=('i686' 'x86_64')
license=('custom')
url="http://www.tortall.net/projects/yasm/"
depends=('glibc')
source=(http://www.tortall.net/projects/yasm/releases/${pkgname}-${pkgver}.tar.gz)
options=('!libtool')
md5sums=('4cfc0686cf5350dd1305c4d905eb55a6')

build() {
  cd ${srcdir}/${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make
}

check() {
  cd ${srcdir}/${pkgname}-${pkgver}
  make check
}

package() {
  cd ${srcdir}/${pkgname}-${pkgver}
  make DESTDIR=${pkgdir} install
  install -Dm644 COPYING ${pkgdir}/usr/share/licenses/${pkgname}/LICENSE
}
